Rules for integrands involving Fresnel integral functions

1. $\int \text{FresnelS}[a+b\,x]^n\,dx$

1: $\int \text{FresnelS}[a+b\,x]\,dx$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{FresnelS}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\text{FresnelS}[a+b\,x]}{b}+\frac{\text{Cos}\left[\frac{\pi}{2}\,(a+b\,x)^2\right]}{b\,\pi}$$

- **Program code:**

```
Int[FresnelS[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*FresnelS[a+b*x]/b + Cos[Pi/2*(a+b*x)^2]/(b*Pi) /;
FreeQ[{a,b},x]
```

```
Int[FresnelC[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*FresnelC[a+b*x]/b - Sin[Pi/2*(a+b*x)^2]/(b*Pi) /;
FreeQ[{a,b},x]
```

2: $\int \text{FresnelS}[a+b\,x]^2\,dx$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{FresnelS}[a+b\,x]^2\,dx \longrightarrow \frac{(a+b\,x)\,\text{FresnelS}[a+b\,x]^2}{b}-2\int (a+b\,x)\,\text{Sin}\left[\frac{\pi}{2}\,(a+b\,x)^2\right]\text{FresnelS}[a+b\,x]\,dx$$

- **Program code:**

```
Int[FresnelS[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*FresnelS[a+b*x]^2/b -
  2*Int[(a+b*x)*Sin[Pi/2*(a+b*x)^2]*FresnelS[a+b*x],x] /;
FreeQ[{a,b},x]
```

```
Int[FresnelC[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*FresnelC[a+b*x]^2/b -
  2*Int[(a+b*x)*Cos[Pi/2*(a+b*x)^2]*FresnelC[a+b*x],x] /;
FreeQ[{a,b},x]
```

X: $\int \text{FresnelS}[a+b\,x]^n\,dx$ when $n \neq 1 \wedge n \neq 2$

- **Rule: If $n \neq 1 \wedge n \neq 2$, then**

$$\int \text{FresnelS}[a+b\,x]^n\,dx \longrightarrow \int \text{FresnelS}[a+b\,x]^n\,dx$$

- **Program code:**

```
Int[FresnelS[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,n},x] && NeQ[n,1] && NeQ[n,2]
```

```
Int[FresnelC[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,n},x] && NeQ[n,1] && NeQ[n,2]
```

2. $\int (c + d\,x)^m\, \text{FresnelS}[a + b\,x]^n\, dx$

   1. $\int (c + d\,x)^m\, \text{FresnelS}[a + b\,x]\, dx$

      1. $\int (d\,x)^m\, \text{FresnelS}[b\,x]\, dx$

         1: $\int \frac{\text{FresnelS}[b\,x]}{x}\, dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\text{FresnelS}[b\,x] == \frac{1+i}{4}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1+i)\,b\,x\right] + \frac{1-i}{4}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1-i)\,b\,x\right]$
- **Basis:** $\text{FresnelC}[b\,x] == \frac{1-i}{4}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1+i)\,b\,x\right] + \frac{1+i}{4}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1-i)\,b\,x\right]$
- **Rule:**

$$\int \frac{\text{FresnelS}[b\,x]}{x}\, dx \longrightarrow \frac{1+i}{4}\int \frac{\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1+i)\,b\,x\right]}{x}\, dx + \frac{1-i}{4}\int \frac{\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1-i)\,b\,x\right]}{x}\, dx$$

- **Program code:**

```
Int[FresnelS[b_.*x_]/x_,x_Symbol] :=
  (1+I)/4*Int[Erf[Sqrt[Pi]/2*(1+I)*b*x]/x,x] + (1-I)/4*Int[Erf[Sqrt[Pi]/2*(1-I)*b*x]/x,x] /;
FreeQ[b,x]
```

```
Int[FresnelC[b_.*x_]/x_,x_Symbol] :=
  (1-I)/4*Int[Erf[Sqrt[Pi]/2*(1+I)*b*x]/x,x] + (1+I)/4*Int[Erf[Sqrt[Pi]/2*(1-I)*b*x]/x,x] /;
FreeQ[b,x]
```

2: $\int (d\,x)^m\,\text{FresnelS}[b\,x]\,dx$ when $m \neq -1$

- **Derivation: Integration by parts**
- **Rule: If $m \neq -1$, then**

$$\int (d\,x)^m\,\text{FresnelS}[b\,x]\,dx \longrightarrow \frac{(d\,x)^{m+1}\,\text{FresnelS}[b\,x]}{d\,(m+1)} - \frac{b}{d\,(m+1)}\int (d\,x)^{m+1}\,\text{Sin}\!\left[\frac{\pi}{2}\,b^2\,x^2\right]dx$$

- **Program code:**

```
Int[(d_.*x_)^m_.*FresnelS[b_.*x_],x_Symbol] :=
  (d*x)^(m+1)*FresnelS[b*x]/(d*(m+1)) - b/(d*(m+1))*Int[(d*x)^(m+1)*Sin[Pi/2*b^2*x^2],x] /;
FreeQ[{b,d,m},x] && NeQ[m,-1]
```

```
Int[(d_.*x_)^m_.*FresnelC[b_.*x_],x_Symbol] :=
  (d*x)^(m+1)*FresnelC[b*x]/(d*(m+1)) - b/(d*(m+1))*Int[(d*x)^(m+1)*Cos[Pi/2*b^2*x^2],x] /;
FreeQ[{b,d,m},x] && NeQ[m,-1]
```

2: $\int (c+d\,x)^m\,\text{FresnelS}[a+b\,x]\,dx$ when $m \in \mathbb{Z}^+$

- **Derivation: Integration by parts**
- **Rule: If $m \in \mathbb{Z}^+$, then**

$$\int (c+d\,x)^m\,\text{FresnelS}[a+b\,x]\,dx \longrightarrow \frac{(c+d\,x)^{m+1}\,\text{FresnelS}[a+b\,x]}{d\,(m+1)} - \frac{b}{d\,(m+1)}\int (c+d\,x)^{m+1}\,\text{Sin}\!\left[\frac{\pi}{2}\,(a+b\,x)^2\right]dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*FresnelS[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*FresnelS[a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*Sin[Pi/2*(a+b*x)^2],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*FresnelC[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*FresnelC[a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*Cos[Pi/2*(a+b*x)^2],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

2. $\int (c+d\,x)^m\,\text{FresnelS}[a+b\,x]^2\,dx$

1: $\int x^m\,\text{FresnelS}[b\,x]^2\,dx$ when $m\in\mathbb{Z}\land m\neq -1$

- **Derivation: Integration by parts**
- **Rule: If $m\in\mathbb{Z}\land m\neq -1$, then**

$$\int x^m\,\text{FresnelS}[b\,x]^2\,dx\,\longrightarrow\,\frac{x^{m+1}\,\text{FresnelS}[b\,x]^2}{m+1}-\frac{2\,b}{m+1}\int x^{m+1}\,\text{Sin}\Big[\frac{\pi}{2}\,b^2\,x^2\Big]\,\text{FresnelS}[b\,x]\,dx$$

- **Program code:**

```
Int[x_^m_.*FresnelS[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*FresnelS[b*x]^2/(m+1) -
  2*b/(m+1)*Int[x^(m+1)*Sin[Pi/2*b^2*x^2]*FresnelS[b*x],x] /;
FreeQ[b,x] && IntegerQ[m] && NeQ[m,-1]
```

```
Int[x_^m_.*FresnelC[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*FresnelC[b*x]^2/(m+1) -
  2*b/(m+1)*Int[x^(m+1)*Cos[Pi/2*b^2*x^2]*FresnelC[b*x],x] /;
FreeQ[b,x] && IntegerQ[m] && NeQ[m,-1]
```

2: $\int (c+d\,x)^m\,\text{FresnelS}[a+b\,x]^2\,dx$ when $m\in\mathbb{Z}^+$

- **Derivation: Integration by substitution**
- **Rule: If $m\in\mathbb{Z}^+$, then**

$$\int (c+d\,x)^m\,\text{FresnelS}[a+b\,x]^2\,dx\,\longrightarrow\,\frac{1}{b^{m+1}}\,\text{Subst}\Big[\int \text{FresnelS}[x]^2\,\text{ExpandIntegrand}[(b\,c-a\,d+d\,x)^m,\,x]\,dx,\,x,\,a+b\,x\Big]$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*FresnelS[a_+b_.*x_]^2,x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[FresnelS[x]^2,(b*c-a*d+d*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*FresnelC[a_+b_.*x_]^2,x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[FresnelC[x]^2,(b*c-a*d+d*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

X: $\int (c+d\,x)^m\,\text{FresnelS}[a+b\,x]^n\,dx$

- **Rule:**

$$\int (c+d\,x)^m\,\text{FresnelS}[a+b\,x]^n\,dx \longrightarrow \int (c+d\,x)^m\,\text{FresnelS}[a+b\,x]^n\,dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*FresnelS[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

```
Int[(c_.+d_.*x_)^m_.*FresnelC[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

3. $\int e^{c+d\,x^2}\,\text{FresnelS}[a+b\,x]^n\,dx$

1: $\int e^{c+d\,x^2}\,\text{FresnelS}[b\,x]\,dx$ when $d^2 == -\frac{\pi^2}{4}\,b^4$

- **Derivation: Algebraic expansion**
- **Basis:** $\text{FresnelS}[b\,x] == \frac{1+i}{4}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1+i)\,b\,x\right] + \frac{1-i}{4}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1-i)\,b\,x\right]$
- **Basis:** $\text{FresnelC}[b\,x] == \frac{1-i}{4}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1+i)\,b\,x\right] + \frac{1+i}{4}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1-i)\,b\,x\right]$
- **Note: If** $d^2 == -\frac{\pi^2}{4}\,b^4$, **then resulting integrands are integrable.**
- **Rule:**

$$\int e^{c+d\,x^2}\,\text{FresnelS}[b\,x]\,dx \longrightarrow \frac{1+i}{4}\int e^{c+d\,x^2}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1+i)\,b\,x\right]dx + \frac{1-i}{4}\int e^{c+d\,x^2}\,\text{Erf}\left[\frac{\sqrt{\pi}}{2}\,(1-i)\,b\,x\right]dx$$

- **Program code:**

```
Int[E^(c_.+d_.*x_^2)*FresnelS[b_.*x_],x_Symbol] :=
  (1+I)/4*Int[E^(c+d*x^2)*Erf[Sqrt[Pi]/2*(1+I)*b*x],x] + (1-I)/4*Int[E^(c+d*x^2)*Erf[Sqrt[Pi]/2*(1-I)*b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-Pi^2/4*b^4]
```

```
Int[E^(c_.+d_.*x_^2)*FresnelC[b_.*x_],x_Symbol] :=
  (1-I)/4*Int[E^(c+d*x^2)*Erf[Sqrt[Pi]/2*(1+I)*b*x],x] + (1+I)/4*Int[E^(c+d*x^2)*Erf[Sqrt[Pi]/2*(1-I)*b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-Pi^2/4*b^4]
```

X: $\int e^{c+d\,x^2}\,\text{FresnelS}[a+b\,x]^n\,dx$

- **Rule:**

$$\int e^{c+d\,x^2}\,\text{FresnelS}[a+b\,x]^n\,dx \longrightarrow \int e^{c+d\,x^2}\,\text{FresnelS}[a+b\,x]^n\,dx$$

- **Program code:**

```
Int[E^(c_.+d_.*x_^2)*FresnelS[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[E^(c+d*x^2)*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[E^(c_.+d_.*x_^2)*FresnelC[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[E^(c+d*x^2)*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

4. $\int \text{Sin}\left[c+d\,x^2\right]\,\text{FresnelS}[a+b\,x]^n\,dx$

1: $\int \text{Sin}\left[d\,x^2\right]\,\text{FresnelS}[b\,x]^n\,dx$ when $d^2 == \frac{\pi^2}{4}\,b^4$

- **Derivation: Integration by substitution**
- **Basis: If** $d^2 == \frac{\pi^2}{4}\,b^4$, **then** $\text{Sin}\left[d\,x^2\right]\,F[\text{FresnelS}[b\,x]] == \frac{\pi\,b}{2\,d}\,\text{Subst}[F[x],\,x,\,\text{FresnelS}[b\,x]]\,\partial_x\text{FresnelS}[b\,x]$
- **Rule: If** $d^2 == \frac{\pi^2}{4}\,b^4$, **then**

$$\int \text{Sin}\left[d\,x^2\right]\,\text{FresnelS}[b\,x]^n\,dx \longrightarrow \frac{\pi\,b}{2\,d}\,\text{Subst}\left[\int x^n\,dx,\,x,\,\text{FresnelS}[b\,x]\right]$$

- **Program code:**

```
Int[Sin[d_.*x_^2]*FresnelS[b_.*x_]^n_.,x_Symbol] :=
  Pi*b/(2*d)*Subst[Int[x^n,x],x,FresnelS[b*x]] /;
FreeQ[{b,d,n},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[Cos[d_.*x_^2]*FresnelC[b_.*x_]^n_.,x_Symbol] :=
  Pi*b/(2*d)*Subst[Int[x^n,x],x,FresnelC[b*x]] /;
FreeQ[{b,d,n},x] && EqQ[d^2,Pi^2/4*b^4]
```

2: $\int \text{Sin}[c + d\,x^2]\ \text{FresnelS}[b\,x]\,dx$ when $d^2 == \frac{\pi^2}{4} b^4$

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sin}[c + d\,x^2] == \text{Sin}[c]\ \text{Cos}[d\,x^2] + \text{Cos}[c]\ \text{Sin}[d\,x^2]$
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4$**, then**

$$\int \text{Sin}[c + d\,x^2]\ \text{FresnelS}[b\,x]\,dx \longrightarrow \text{Sin}[c] \int \text{Cos}[d\,x^2]\ \text{FresnelS}[b\,x]\,dx + \text{Cos}[c] \int \text{Sin}[d\,x^2]\ \text{FresnelS}[b\,x]\,dx$$

- **Program code:**

```
Int[Sin[c_+d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  Sin[c]*Int[Cos[d*x^2]*FresnelS[b*x],x] + Cos[c]*Int[Sin[d*x^2]*FresnelS[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[Cos[c_+d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  Cos[c]*Int[Cos[d*x^2]*FresnelC[b*x],x] - Sin[c]*Int[Sin[d*x^2]*FresnelC[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

X: $\int \text{Sin}[c + d\,x^2]\ \text{FresnelS}[a + b\,x]^n\,dx$

- **Rule:**

$$\int \text{Sin}[c + d\,x^2]\ \text{FresnelS}[a + b\,x]^n\,dx \longrightarrow \int \text{Sin}[c + d\,x^2]\ \text{FresnelS}[a + b\,x]^n\,dx$$

- **Program code:**

```
Int[Sin[c_.+d_.*x_^2]*FresnelS[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[Sin[c+d*x^2]*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[Cos[c_.+d_.*x_^2]*FresnelC[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[Cos[c+d*x^2]*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

5. $\int \text{Cos}\left[c+d\,x^2\right]\,\text{FresnelS}[a+b\,x]^n\,dx$

1: $\int \text{Cos}\left[d\,x^2\right]\,\text{FresnelS}[b\,x]\,dx$ when $d^2 == \frac{\pi^2}{4}\,b^4$

- **Derivation: Algebraic expansion**

- **Rule: If** $d^2 == \frac{\pi^2}{4}\,b^4$**, then**

$$\int \text{Cos}\left[d\,x^2\right]\,\text{FresnelS}[b\,x]\,dx \longrightarrow$$

$$\frac{\text{FresnelC}[b\,x]\,\text{FresnelS}[b\,x]}{2\,b}-\frac{1}{8}\,i\,b\,x^2\,\text{HypergeometricPFQ}\left[\{1,1\},\left\{\frac{3}{2},2\right\},-\frac{1}{2}\,i\,b^2\,\pi\,x^2\right]+\frac{1}{8}\,i\,b\,x^2\,\text{HypergeometricPFQ}\left[\{1,1\},\left\{\frac{3}{2},2\right\},\frac{1}{2}\,i\,b^2\,\pi\,x^2\right]$$

- **Program code:**

```
Int[Cos[d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  FresnelC[b*x]*FresnelS[b*x]/(2*b) -
  1/8*I*b*x^2*HypergeometricPFQ[{1,1},{3/2,2},-1/2*I*b^2*Pi*x^2] +
  1/8*I*b*x^2*HypergeometricPFQ[{1,1},{3/2,2},1/2*I*b^2*Pi*x^2] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[Sin[d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  b*Pi*FresnelC[b*x]*FresnelS[b*x]/(4*d) +
  1/8*I*b*x^2*HypergeometricPFQ[{1,1},{3/2,2},-I*d*x^2] -
  1/8*I*b*x^2*HypergeometricPFQ[{1,1},{3/2,2},I*d*x^2] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

2: $\int \text{Cos}\left[c + d\,x^2\right]\,\text{FresnelS}[b\,x]\,dx$ when $d^2 == \frac{\pi^2}{4}\,b^4$

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Cos}\left[c + d\,x^2\right] == \text{Cos}[c]\,\text{Cos}\left[d\,x^2\right] - \text{Sin}[c]\,\text{Sin}\left[d\,x^2\right]$
- **Rule: If** $d^2 == \frac{\pi^2}{4}\,b^4$**, then**

$$\int \text{Cos}\left[c + d\,x^2\right]\,\text{FresnelS}[b\,x]\,dx \longrightarrow \text{Cos}[c]\int \text{Cos}\left[d\,x^2\right]\,\text{FresnelS}[b\,x]\,dx - \text{Sin}[c]\int \text{Sin}\left[d\,x^2\right]\,\text{FresnelS}[b\,x]\,dx$$

- **Program code:**

```
Int[Cos[c_+d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  Cos[c]*Int[Cos[d*x^2]*FresnelS[b*x],x] - Sin[c]*Int[Sin[d*x^2]*FresnelS[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[Sin[c_+d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  Sin[c]*Int[Cos[d*x^2]*FresnelC[b*x],x] + Cos[c]*Int[Sin[d*x^2]*FresnelC[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

X: $\int \text{Cos}\left[c + d\,x^2\right]\,\text{FresnelS}[a + b\,x]^n\,dx$

- **Rule:**

$$\int \text{Cos}\left[c + d\,x^2\right]\,\text{FresnelS}[a + b\,x]^n\,dx \longrightarrow \int \text{Cos}\left[c + d\,x^2\right]\,\text{FresnelS}[a + b\,x]^n\,dx$$

- **Program code:**

```
Int[Cos[c_.+d_.*x_^2]*FresnelS[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[Cos[c+d*x^2]*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[Sin[c_.+d_.*x_^2]*FresnelC[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[Sin[c+d*x^2]*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

6. $\int (e\,x)^m \operatorname{Sin}[c+d\,x^2]\,\operatorname{FresnelS}[a+b\,x]^n\,dx$

1. $\int x^m \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx$ when $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m \in \mathbb{Z}$

1. $\int x^m \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx$ when $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m \in \mathbb{Z}^+$

1: $\int x \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx$ when $d^2 == \frac{\pi^2}{4} b^4$

- **Derivation: Integration by parts and algebraic simplification**
- **Basis:** $-\partial_x \frac{\operatorname{Cos}[d\,x^2]}{2\,d} == x \operatorname{Sin}[d\,x^2]$
- **Basis: If** $d^2 == \frac{\pi^2}{4} b^4$, **then** $\operatorname{Cos}[d\,x^2] \operatorname{Sin}\left[\frac{1}{2} b^2 \pi\,x^2\right] == \frac{d}{b^2 \pi} \operatorname{Sin}[2\,d\,x^2]$
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4$, **then**

$$\int x \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx \longrightarrow -\frac{\operatorname{Cos}[d\,x^2]\,\operatorname{FresnelS}[b\,x]}{2\,d} + \frac{1}{2\,b\,\mathrm{Pi}} \int \operatorname{Sin}[2\,d\,x^2]\,dx$$

- **Basis:** $\partial_x \frac{\operatorname{Sin}[d\,x^2]}{2\,d} == x \operatorname{Cos}[d\,x^2]$
- **Basis: If** $d^2 == \frac{\pi^2}{4} b^4$, **then** $\operatorname{Sin}[d\,x^2] \operatorname{Cos}\left[\frac{1}{2} b^2 \pi\,x^2\right] == \frac{1}{2} \operatorname{Sin}[2\,d\,x^2]$
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4$, **then**

$$\int x \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelC}[b\,x]\,dx \longrightarrow \frac{\operatorname{Sin}[d\,x^2]\,\operatorname{FresnelC}[b\,x]}{2\,d} - \frac{b}{4\,d} \int \operatorname{Sin}[2\,d\,x^2]\,dx$$

- **Program code:**

```
Int[x_*Sin[d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  -Cos[d*x^2]*FresnelS[b*x]/(2*d) + 1/(2*b*Pi)*Int[Sin[2*d*x^2],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[x_*Cos[d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  Sin[d*x^2]*FresnelC[b*x]/(2*d) - b/(4*d)*Int[Sin[2*d*x^2],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

2: $\int x^m \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx$ when $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m-1 \in \mathbb{Z}^+$

- **Derivation: Integration by parts and algebraic simplification**
- **Basis:** $-\partial_x \frac{\operatorname{Cos}[d\,x^2]}{2\,d} == x \operatorname{Sin}[d\,x^2]$
- **Basis: If** $d^2 == \frac{\pi^2}{4} b^4$, **then** $\operatorname{Cos}[d\,x^2] \operatorname{Sin}\left[\frac{1}{2} b^2 \pi x^2\right] == \frac{d}{b^2 \pi} \operatorname{Sin}[2\,d\,x^2]$
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m-1 \in \mathbb{Z}^+$, **then**

$$\int x^m \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx \longrightarrow$$
$$-\frac{x^{m-1} \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelS}[b\,x]}{2\,d} + \frac{1}{2\,b\,\mathrm{Pi}} \int x^{m-1} \operatorname{Sin}[2\,d\,x^2]\,dx + \frac{m-1}{2\,d} \int x^{m-2} \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx$$

- **Basis:** $\partial_x \frac{\operatorname{Sin}[d\,x^2]}{2\,d} == x \operatorname{Cos}[d\,x^2]$
- **Basis: If** $d^2 == \frac{\pi^2}{4} b^4$, **then** $\operatorname{Sin}[d\,x^2] \operatorname{Cos}\left[\frac{1}{2} b^2 \pi x^2\right] == \frac{1}{2} \operatorname{Sin}[2\,d\,x^2]$
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m-1 \in \mathbb{Z}^+$, **then**

$$\int x^m \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelC}[b\,x]\,dx \longrightarrow$$
$$\frac{x^{m-1} \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelC}[b\,x]}{2\,d} - \frac{b}{4\,d} \int x^{m-1} \operatorname{Sin}[2\,d\,x^2]\,dx - \frac{m-1}{2\,d} \int x^{m-2} \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelC}[b\,x]\,dx$$

- **Program code:**

```
Int[x_^m_*Sin[d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  -x^(m-1)*Cos[d*x^2]*FresnelS[b*x]/(2*d) +
  1/(2*b*Pi)*Int[x^(m-1)*Sin[2*d*x^2],x] +
  (m-1)/(2*d)*Int[x^(m-2)*Cos[d*x^2]*FresnelS[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && IGtQ[m,1]
```

```
Int[x_^m_*Cos[d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  x^(m-1)*Sin[d*x^2]*FresnelC[b*x]/(2*d) -
  b/(4*d)*Int[x^(m-1)*Sin[2*d*x^2],x] -
  (m-1)/(2*d)*Int[x^(m-2)*Sin[d*x^2]*FresnelC[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && IGtQ[m,1]
```

2: $\int x^m \operatorname{Sin}\left[d\, x^2\right] \operatorname{FresnelS}[b\, x]\, dx$ when $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m+2 \in \mathbb{Z}^-$

- **Derivation: Inverted integration by parts**
- **Rule: If $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m+2 \in \mathbb{Z}^-$, then**

$$\int x^m \operatorname{Sin}\left[d\, x^2\right] \operatorname{FresnelS}[b\, x]\, dx \longrightarrow$$

$$\frac{x^{m+1} \operatorname{Sin}\left[d\, x^2\right] \operatorname{FresnelS}[b\, x]}{m+1} - \frac{d\, x^{m+2}}{\pi\, b\, (m+1)\, (m+2)} + \frac{d}{\pi\, b\, (m+1)} \int x^{m+1} \operatorname{Cos}\left[2\, d\, x^2\right] dx - \frac{2\, d}{m+1} \int x^{m+2} \operatorname{Cos}\left[d\, x^2\right] \operatorname{FresnelS}[b\, x]\, dx$$

$$\int x^m \operatorname{Cos}\left[d\, x^2\right] \operatorname{FresnelC}[b\, x]\, dx \longrightarrow$$

$$\frac{x^{m+1} \operatorname{Cos}\left[d\, x^2\right] \operatorname{FresnelC}[b\, x]}{m+1} - \frac{b\, x^{m+2}}{2\, (m+1)\, (m+2)} - \frac{b}{2\, (m+1)} \int x^{m+1} \operatorname{Cos}\left[2\, d\, x^2\right] dx + \frac{2\, d}{m+1} \int x^{m+2} \operatorname{Sin}\left[d\, x^2\right] \operatorname{FresnelC}[b\, x]\, dx$$

- **Program code:**

```
Int[x_^m_*Sin[d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  x^(m+1)*Sin[d*x^2]*FresnelS[b*x]/(m+1) -
  d*x^(m+2)/(Pi*b*(m+1)*(m+2)) +
  d/(Pi*b*(m+1))*Int[x^(m+1)*Cos[2*d*x^2],x] -
  2*d/(m+1)*Int[x^(m+2)*Cos[d*x^2]*FresnelS[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && ILtQ[m,-2]
```

```
Int[x_^m_*Cos[d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  x^(m+1)*Cos[d*x^2]*FresnelC[b*x]/(m+1) -
  b*x^(m+2)/(2*(m+1)*(m+2)) -
  b/(2*(m+1))*Int[x^(m+1)*Cos[2*d*x^2],x] +
  2*d/(m+1)*Int[x^(m+2)*Sin[d*x^2]*FresnelC[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && ILtQ[m,-2]
```

X: $\int (e\,x)^m \operatorname{Sin}\left[c+d\,x^2\right] \operatorname{FresnelS}[a+b\,x]^n \,dx$

- **Rule:**

$$\int (e\,x)^m \operatorname{Sin}\left[c+d\,x^2\right] \operatorname{FresnelS}[a+b\,x]^n \,dx \longrightarrow \int (e\,x)^m \operatorname{Sin}\left[c+d\,x^2\right] \operatorname{FresnelS}[a+b\,x]^n \,dx$$

- **Program code:**

```
Int[(e_.*x_)^m_.*Sin[c_.+d_.*x_^2]*FresnelS[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(e*x)^m*Sin[c+d*x^2]*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*Cos[c_.+d_.*x_^2]*FresnelC[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(e*x)^m*Cos[c+d*x^2]*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

7. $\int (e\,x)^m \operatorname{Cos}\left[c+d\,x^2\right] \operatorname{FresnelS}[a+b\,x]^n \,dx$

1. $\int x^m \operatorname{Cos}\left[d\,x^2\right] \operatorname{FresnelS}[b\,x] \,dx$ when $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m \in \mathbb{Z}$

1. $\int x^m \operatorname{Cos}\left[d\,x^2\right] \operatorname{FresnelS}[b\,x] \,dx$ when $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m \in \mathbb{Z}^+$

1: $\int x \operatorname{Cos}\left[d\,x^2\right] \operatorname{FresnelS}[b\,x] \,dx$ when $d^2 == \frac{\pi^2}{4} b^4$

- **Derivation: Integration by parts and algebraic simplification**
- **Basis:** $\partial_x \frac{\operatorname{Sin}[d\,x^2]}{2\,d} == x \operatorname{Cos}\left[d\,x^2\right]$
- **Basis: If** $d^2 == \frac{\pi^2}{4} b^4$**, then** $\operatorname{Sin}\left[d\,x^2\right] \operatorname{Sin}\left[\frac{1}{2} b^2 \pi\,x^2\right] == \frac{2\,d}{\pi\,b^2} \operatorname{Sin}\left[d\,x^2\right]^2$
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4$**, then**

$$\int x \operatorname{Cos}\left[d\,x^2\right] \operatorname{FresnelS}[b\,x] \,dx \longrightarrow \frac{\operatorname{Sin}\left[d\,x^2\right] \operatorname{FresnelS}[b\,x]}{2\,d} - \frac{1}{\pi\,b} \int \operatorname{Sin}\left[d\,x^2\right]^2 \,dx$$

- **Basis:** $-\partial_x \frac{\operatorname{Cos}[d\,x^2]}{2\,d} == x \operatorname{Sin}\left[d\,x^2\right]$
- **Basis: If** $d^2 == \frac{\pi^2}{4} b^4$**, then** $\operatorname{Cos}\left[d\,x^2\right] \operatorname{Cos}\left[\frac{1}{2} b^2 \pi\,x^2\right] == \operatorname{Cos}\left[d\,x^2\right]^2$
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4$**, then**

$$\int x\,\text{Sin}\left[d\,x^2\right]\,\text{FresnelC}[b\,x]\,dx \longrightarrow -\frac{\text{Cos}\left[d\,x^2\right]\,\text{FresnelC}[b\,x]}{2\,d}+\frac{b}{2\,d}\int\text{Cos}\left[d\,x^2\right]^2\,dx$$

- **Program code:**

```
Int[x_*Cos[d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  Sin[d*x^2]*FresnelS[b*x]/(2*d) - 1/(Pi*b)*Int[Sin[d*x^2]^2,x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

```
Int[x_*Sin[d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  -Cos[d*x^2]*FresnelC[b*x]/(2*d) + b/(2*d)*Int[Cos[d*x^2]^2,x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4]
```

2: $\int x^m \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx$ when $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m-1 \in \mathbb{Z}^+$

- **Derivation: Integration by parts and algebraic simplification**
- **Basis:** $\partial_x \frac{\operatorname{Sin}[d\,x^2]}{2\,d} == x \operatorname{Cos}[d\,x^2]$
- **Basis: If** $d^2 == \frac{\pi^2}{4} b^4$**, then** $\operatorname{Sin}[d\,x^2] \operatorname{Sin}\left[\frac{1}{2} b^2 \pi x^2\right] == \frac{2\,d}{\pi\,b^2} \operatorname{Sin}[d\,x^2]^2$
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m-1 \in \mathbb{Z}^+$**, then**

$$\int x^m \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx \longrightarrow$$
$$\frac{x^{m-1} \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelS}[b\,x]}{2\,d} - \frac{1}{\pi\,b} \int x^{m-1} \operatorname{Sin}[d\,x^2]^2\,dx - \frac{m-1}{2\,d} \int x^{m-2} \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx$$

- **Basis:** $-\partial_x \frac{\operatorname{Cos}[d\,x^2]}{2\,d} == x \operatorname{Sin}[d\,x^2]$
- **Basis: If** $d^2 == \frac{\pi^2}{4} b^4$**, then** $\operatorname{Cos}[d\,x^2] \operatorname{Cos}\left[\frac{1}{2} b^2 \pi x^2\right] == \operatorname{Cos}[d\,x^2]^2$
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4$**, then**

$$\int x^m \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelC}[b\,x]\,dx \longrightarrow$$
$$-\frac{x^{m-1} \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelC}[b\,x]}{2\,d} + \frac{b}{2\,d} \int x^{m-1} \operatorname{Cos}[d\,x^2]^2\,dx + \frac{m-1}{2\,d} \int x^{m-2} \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelC}[b\,x]\,dx$$

- **Program code:**

```
Int[x_^m_*Cos[d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  x^(m-1)*Sin[d*x^2]*FresnelS[b*x]/(2*d) -
  1/(Pi*b)*Int[x^(m-1)*Sin[d*x^2]^2,x] -
  (m-1)/(2*d)*Int[x^(m-2)*Sin[d*x^2]*FresnelS[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && IGtQ[m,1]
```

```
Int[x_^m_*Sin[d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  -x^(m-1)*Cos[d*x^2]*FresnelC[b*x]/(2*d) +
  b/(2*d)*Int[x^(m-1)*Cos[d*x^2]^2,x] +
  (m-1)/(2*d)*Int[x^(m-2)*Cos[d*x^2]*FresnelC[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && IGtQ[m,1]
```

2: $\int x^m \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx$ when $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m+1 \in \mathbb{Z}^-$

- **Derivation: Inverted integration by parts**
- **Rule: If** $d^2 == \frac{\pi^2}{4} b^4 \bigwedge m+1 \in \mathbb{Z}^-$**, then**

$$\int x^m \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx \longrightarrow$$
$$\frac{x^{m+1} \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelS}[b\,x]}{m+1} - \frac{d}{\pi\,b\,(m+1)} \int x^{m+1} \operatorname{Sin}[2\,d\,x^2]\,dx + \frac{2\,d}{m+1} \int x^{m+2} \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelS}[b\,x]\,dx$$

$$\int x^m \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelC}[b\,x]\,dx \longrightarrow$$
$$\frac{x^{m+1} \operatorname{Sin}[d\,x^2]\,\operatorname{FresnelC}[b\,x]}{m+1} - \frac{b}{2\,(m+1)} \int x^{m+1} \operatorname{Sin}[2\,d\,x^2]\,dx - \frac{2\,d}{m+1} \int x^{m+2} \operatorname{Cos}[d\,x^2]\,\operatorname{FresnelC}[b\,x]\,dx$$

- **Program code:**

```
Int[x_^m_*Cos[d_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  x^(m+1)*Cos[d*x^2]*FresnelS[b*x]/(m+1) -
  d/(Pi*b*(m+1))*Int[x^(m+1)*Sin[2*d*x^2],x] +
  2*d/(m+1)*Int[x^(m+2)*Sin[d*x^2]*FresnelS[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && ILtQ[m,-1]
```

```
Int[x_^m_*Sin[d_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  x^(m+1)*Sin[d*x^2]*FresnelC[b*x]/(m+1) -
  b/(2*(m+1))*Int[x^(m+1)*Sin[2*d*x^2],x] -
  2*d/(m+1)*Int[x^(m+2)*Cos[d*x^2]*FresnelC[b*x],x] /;
FreeQ[{b,d},x] && EqQ[d^2,Pi^2/4*b^4] && ILtQ[m,-1]
```

X: $\int (e\,x)^m \operatorname{Cos}\left[c+d\,x^2\right] \operatorname{FresnelS}[a+b\,x]^n \,dx$

- **Rule:**

$$\int (e\,x)^m \operatorname{Cos}\left[c+d\,x^2\right] \operatorname{FresnelS}[a+b\,x]^n \,dx \longrightarrow \int (e\,x)^m \operatorname{Cos}\left[c+d\,x^2\right] \operatorname{FresnelS}[a+b\,x]^n \,dx$$

- **Program code:**

```
Int[(e_.*x_)^m_.*Cos[c_.+d_.*x_^2]*FresnelS[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(e*x)^m*Cos[c+d*x^2]*FresnelS[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*Sin[c_.+d_.*x_^2]*FresnelC[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(e*x)^m*Sin[c+d*x^2]*FresnelC[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```